/////// TABLE 1 ////////// 

use "~/Dropbox/Replication_MVC/Datasets/datasets_analysis/panel_excombatientes.dta", clear 

******************************************** 
*** TABLE 1 - Panle A (\eta and \theta)  ***
********************************************
eststo clear
set more off

* Main - Red-handed Captures 
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r, absorb(wartimenetwork year origmun) vce(cluster wartimenetwork)
estadd local ALLFE = "$\checkmark$"
summ diffcaptures_f if e(sample) ==1
estadd local meanDep = trim("`: display %12.4fc `r(mean)''")
estadd local SDDep = trim("`: display %12.4fc `r(sd)''")
eststo

*  + Time trends
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r time_t, absorb(wartimenetwork origmun year) vce(cluster wartimenetwork)
estadd local ALLFE = "$\checkmark$"
estadd local TT = "$\checkmark$"
summ diffcaptures_f if e(sample) ==1
estadd local meanDep = trim("`: display %12.4fc `r(mean)''")
estadd local SDDep = trim("`: display %12.4fc `r(sd)''")
eststo

*  + Region * Time 
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r, absorb(wartimenetwork year origmun c.time_t##i.region) vce(cluster wartimenetwork)
estadd local ALLFE = "$\checkmark$"
estadd local TT = "$\checkmark$"
estadd local MT = "$\checkmark$"
summ diffcaptures_f if e(sample) ==1
estadd local meanDep = trim("`: display %12.4fc `r(mean)''")
estadd local SDDep = trim("`: display %12.4fc `r(sd)''")
eststo

*  + Municipality baseline covariate intercated with year
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r c.pobreza##c.time_t c.pobl_tot##c.time_t c.disbogota##c.time_t c.paved_roads##c.time_t , absorb(wartimenetwork year origmun c.time_t##i.region) vce(cluster wartimenetwork)
estadd local ALLFE = "$\checkmark$"
estadd local TT = "$\checkmark$"
estadd local MT = "$\checkmark$"
estadd local BC = "$\checkmark$"
summ diffcaptures_f if e(sample) ==1
estadd local meanDep = trim("`: display %12.4fc `r(mean)''")
estadd local SDDep = trim("`: display %12.4fc `r(sd)''")
eststo

* + Ex-cobatants baseline covariates
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r age female black indigenous c.pobreza##c.time_t c.pobl_tot##c.time_t c.disbogota##c.time_t c.paved_roads##c.time_t, absorb(wartimenetwork year origmun c.time_t##i.region) vce(cluster wartimenetwork)
estadd local ALLFE = "$\checkmark$"
estadd local TT = "$\checkmark$"
estadd local MT = "$\checkmark$"
estadd local BC = "$\checkmark$"
estadd local CO = "$\checkmark$"
summ diffcaptures_f if e(sample) ==1
estadd local meanDep = trim("`: display %12.4fc `r(mean)''")
estadd local SDDep = trim("`: display %12.4fc `r(sd)''")
eststo

esttab est* using "~/Dropbox/Replication_MVC/Paper_tables/Table1_panelA.tex", se ///
	   star(* 0.10 ** 0.05 *** 0.01 ) ///
	   stats(meanDep SDDep N ALLFE TT MT BC CO, fmt(%12.4fc %12.4fc  %12.0fc %12.0fc %12.0fc %12.0fc %12.0fc %12.0fc) ///
	 label("\small Mean of Outcome" "\small S.D of Outcome" "\small Observations" "\small Municipality, Year, and Group FE" "\small Time Trends" "\small Region $\times$  Year" "\small Municipality Characteristics TT" ///
			"\small Individual Covariates" )) ///
	   keep(gold_shock_r mean_gold_shock_r  ) ///
	   varlabels(gold_shock_r "Economic Shock" mean_gold_shock_r "Average Shock" ) ///
	   ti("Economic Shock and Peer Effects for fragrant captures") ///	  
	   replace

	   
*******************************************	   
************* TABLE 1 - Panel B (\beta) 
*******************************************

eststo clear
set more off
* Main - Captures 
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r, absorb(wartimenetwork year origmun) vce(cluster wartimenetwork)
nlcom _b[mean_gold_shock_r]/(_b[gold_shock_r] + _b[mean_gold_shock_r]), post
eststo

*  + Time trends
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r time_t, absorb(wartimenetwork origmun year) vce(cluster wartimenetwork)
nlcom _b[mean_gold_shock_r]/(_b[gold_shock_r] + _b[mean_gold_shock_r]), post
eststo

*  + Region * Time
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r, absorb(wartimenetwork year origmun c.time_t##i.region) vce(cluster wartimenetwork)
nlcom _b[mean_gold_shock_r]/(_b[gold_shock_r] + _b[mean_gold_shock_r]), post
eststo

*  + Municipality baseline covariate intercated with year
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r c.pobreza##c.time_t c.pobl_tot##c.time_t c.disbogota##c.time_t c.paved_roads##c.time_t , absorb(wartimenetwork year origmun c.time_t##i.region) vce(cluster wartimenetwork)
nlcom _b[mean_gold_shock_r]/(_b[gold_shock_r] + _b[mean_gold_shock_r]), post
eststo

* + Ex-cobatants baseline covariates
reghdfe diffcaptures_f gold_shock_r mean_gold_shock_r age female black indigenous c.pobreza##c.time_t c.pobl_tot##c.time_t c.disbogota##c.time_t c.paved_roads##c.time_t, absorb(wartimenetwork year origmun c.time_t##i.region ) vce(cluster wartimenetwork)
nlcom _b[mean_gold_shock_r]/(_b[gold_shock_r] + _b[mean_gold_shock_r]), post
eststo

esttab est* using "~/Dropbox/Replication_MVC/Paper_tables/Table1_panelB.tex", se ///
	   star(* 0.10 ** 0.05 *** 0.01 ) ///
	   stats(meanDep N, fmt(%12.4fc %12.0fc)) ///
	   keep(_nl_1 ) ///
	   varlabels( _nl_1 "Peer Effect") ///
	   ti("Economic Shock and Peer Effects") ///	  
	   replace	   
	   
